
Pràctica 4 1

Pràctiques integrades

Pràctica 4

Matemàtiques, curs 2001-2002.

4 Resoldre equacions

A aquesta pràctica aprendreu a aplicar la comanda de Maple solve( ) per a obtenir les solucions exactes
d’equacions (quan això sigui possible). Recordeu que en nombrosos casos no és possible obtenir solucions
exactes de les equacions i que s’ha de deixar la feina als solucionadors numèrics per a poder calcular solucions
aproximades. Més endavant en aquesta mateixa secció utilitzareu la comanda de Maple fsolve( ) per a
obtenir aproximacions decimals de solucions. També es discutirà la resolució de sistemes d’equacions lineals.
Comencem reinicialitzant les variables i carregant el paquet plots.

> restart:

> with(plots):

4.1 Introduir i manipular equacions: les comandes lhs( ) i rhs( )

Exemple 1:
Es pot donar un nom a una equació igual que es fa amb qualsevol altre expressió. En la ĺınia següent
introdüım l’equació x3 − 5x2 + 23 = 2x2 + 4x− 8 i li posem com a nom eqn1.

> eqn1:=x^3-5*x^2+23=2*x^2+4*x-8;

Exemple 2:
Podem äıllar la part esquerra i la part dreta de l’equació utilitzant les comandes lhs( ) i rhs( ).

> lhs(eqn1);

> rhs(eqn1);

Exemple 3:
Utilitzem les comandes lhs( ) i rhs( ) per a obtenir una equació que és equivalent a l’equació original eqn1
però que té com a part dreta un 0. Posem-li com etiqueta eqn2.

> eqn2:=lhs(eqn1)-rhs(eqn1)=0;

4.2 Obtenir solucions exactes: la comanda solve( )

Primer considerem equacions polinòmiques. Existeixen algoritmes per a calcular les solucions exactes per a
polinomis que tenen fins a grau 4. La comanda de Maple solve( ) coneix aquests algoritmes.

Exemple 1:
Per a obtenir les solucions exactes de l’equació polinomial 3x3 − 4x2 − 43x+ 84 = 0 utilitzem la comanda
solve( ). Noteu que el segon argument de la comanda li diu a Maple que x és la incògnita respecte de la
que volem obtenir la solució.

> solve(3*x^3-4*x^2-43*x+84=0,x);

Aqúı Maple ha trobat les tres solucions i les ha ensenyades.

Exemple 2:
A vegades voldrem seleccionar una solució de la llista de solucions i utilitzar-la en un altre càlcul. Podem
fer això assignant primer un nom (en aquest cas utilitzem la lletra N) al resultat de la comanda solve().
Aleshores N[1] és el primer nombre de la llista, N[2] és el segon i aix́ı successivament (fixeu-vos en els
claudàtors).

> N:=solve(x^2-5*x+3=0,x);

> N[1];

Exemple 3:
Quan es treballa amb la comanda solve( ) sovint és convenient començar donant un nom a l’equació. Noteu
que utilitzem “:=” per assignar el nom i només “=” per a l’equació pròpiament dita.

> eqn1:=7*x^3-11*x^2-27*x-9=0;

Seguidament resolem l’equació utilitzant la comanda solve( ) i assignem el nom H al resultat.

> H:=solve(eqn1,x);

Per a practicar comproveu que cada un d’aquests valors satisfà l’equació. Això es pot fer fàcilment utilitzant
la comanda subs( ).

> subs(x=H[1],eqn1);

> subs(x=H[2],eqn1);

> subs(x=H[3],eqn1);

Exemple 4:
A vegades les solucions “exactes” són massa complicades per a poder ser realment útils. En les dues ĺınies
següents resolem l’equació x3 − 34x2 + 4 = 0.

> eqn1:=x^3-34*x^2+4=0;

> H:=solve(eqn1,x);

Com podreu veure, llegir aquestes solucions exactes és realment un repte! Noteu que I representa
√
−1.

Quan una solució és aix́ı de complicada és més útil mirar una solució aproximada utilitzant evalf( ).

> evalf(H);

Una bona alternativa a la comanda solve( ) en una situació d’aquest tipus és la comada fsolve( ) que
discutirem en la secció següent.
La comanda solve( ) també es pot utilitzar per a determinar solucions exactes d’equacions no polinòmiques.
Hi ha una llista d’alguns exemples simples a baix. En tot cas, si les equacions són complicades, per exemple
si combinen exponencials, polinomis i expressions trigonomètriques, normalment no es podrà disposar de les
solucions exactes. Un altre cop la comanda fsolve( ) és una alternativa.

Exemple 5:
Resoleu l’equació: 5 e(x

4
) = 43

> solve(5*exp(x/4)=43,x);

Exemple 6:
A vegades Maple no mostra totes les solucions possibles. Com podrieu utilitzar el resultat d’aqúı sota per
a poder escriure el conjunt de totes les solucions de l’equació?

> solve(sin(x)=1/2,x);

Exercici 4.1

Resoleu l’equació x3 − 11x2 + 7x+ 147 = 0. Per què Maple produeix només dues solucions diferents per a
aquesta equació cúbica? Per què una d’elles està escrita dues vegades? (Indicació: Factoritzeu la part
esquerra de l’equació).

El fet que (x− 7) és un factor repetit porta com a conseqüència que l’equació cúbica tingu només dues
solucions diferents −3 i 7. Diem que l’arrel 7 té multiplicitat 2, això significa aqúı que hi ha dos factors de
la forma (x− 7) en la factorització del polinomi.

4.3 Obtenir solucions aproximades: la comanda fsolve( )

Es pot utilitzar per a aproximar solucions de qualsevol tipus d’equació. Per a equacions polinòmiques
fsolve() produeix una llista completa de totes les solucions reals en un sol pas (mireu l’Exemple 1). Per
a altres tipus d’equacions, fsolve( ) es pot utilitzar per a obtenir les solucions d’una en una (mireu els
Exemples 2 i 3).

Exemple 1:
La comanda de Maple fsolve( ) calcularà una aproximació numèrica per a cada una de les solucions reals
d’una equació polinòmica. Aproximem totes les solucions reals de l’equació: x4 − x3 − 17x2 − 6x+ 2 = 0.

> eqn:=x^4-x^3-17*x^2-6*x+2=0;

> fsolve(eqn,x);

Les quatre solucions que es mostren a sota ens donen una llista completa de les solucions de l’equació
polinòmica.

Exemple 2:
Trobem totes les solucions reals de l’equació x3 + 1− ex = 0 utilitzant la comanda fsolve( ).

> eqn:=x^3+1-exp(x)=0;

> fsolve(eqn,x);

Maple ens dóna una solució real. Aquest cop Maple no ens ha explicat la historia completa. Hi ha altres
solucions? Com es poden trobar? En l’Exemple 3 es presenta un procediment sistemàtic per a determinar
les solucions que resten.

Exemple 3:
Trobem les altres solucions reals de l’equació x3 + 1− ex = 0 .

El primer pas per a trobar les altres solucions és fer un dibuix del gràfic de la part esquerra de l’equació.
Nota: Recordeu que els talls amb l’eix de les x de y = x3 + 1− ex es corresponen exactament amb les
solucions de l’equació x3 + 1− ex = 0.

> plot(x^3+1-exp(x),x=-3..5,y=-5..15);

El gràfic mostra quatre interseccions amb l’eix de les x. Una d’elles correspon a la solució que hem obtingut
en l’Exemple 2. Quina? Com trobarieu les altres que falten?

Podem extendre la comanda fsolve( ) per a que miri de trobar una solució en un interval particular. Per
exemple per a trobar la solució negativa li demanem a Maple que busqui en l’interval [−1, −0.2] ja que podem
veure a partir del gràfic que hi ha exactament una solució en aquest interval.

> fsolve(eqn,x=-1..-.2);

Per a determinar les altres dues solucions utilitzem fsolve( ) un altre cop, aquesta vegada buscant a
l’interval [1, 2] i a l’interval [4, 5].

> fsolve(eqn,x=1..2);
> fsolve(eqn,x=4..5);

Què passa si demaneu a Maple que busqui una solució en un interval on no hi ha solucions?
Proveu-ho. A partir del gràfic és clar que no hi ha talls amb l’eix de les x (i per tant no hi ha solucions)
entre 2 i 4.

> fsolve(eqn,x=2..4);

Noteu que Maple simplement respon amb la ĺınia que originalment heu introdüıt sense cap canvi quan no
pot trobar una solució a l’interval donat.

Hi ha altres solucions? Per exemple, hi ha alguna solució per a x més gran que 5? Podem comprovar això
fent més gran l’interval sobre el que fem el gràfic. En la ĺınia següent allarguem l’interval fins al [−3, 50].
No apareixen nous talls amb l’eix de les x. El gràfic confirma el que podem esperar mirant els termes de
l’equació, és a dir el terme exponencial domina i fa que el gràfic vagi baixant a la llarga.

> plot(x^3+1-exp(x),x=-3..50,y=-10..15);

Alternativament podem utilitzar la comanda fsolve( ), ara buscant en aquest interval més gran.
> fsolve(eqn,x=5..50);

Tal com esperàvem Maple no haurà trobat solucions.
D’una forma semblant podem comprovar si hi ha solucions cap a l’esquerra. Aqúı busquem si hi ha
solucions a l’interval [−50, −1].

> fsolve(eqn,x=-50..-1);

Cap ni una tampoc!
Ara tenim una llista completa de les solucions aproximades de la nostra equació original x3 + 1− ex = 0 .
Són: −.8251554597, 0, 1.545007279 i 4.567036837.
Exemple 4:
Utilitzem fsolve( ) per a calcular les solucions aproximades de l’equació: x2

20 − 10x = 15 cos(x+ 15).
Com en l’últim exemple utilitzarem un gràfic per ajudar-nos a determinar el nombre i la situació aproximada
de les solucions. La nostra feina se simplifica si comencem convertint l’equació que tenim en una d’equivalent
que té com a part dreta un 0. Aix́ı resoldrem l’equació equivalent: x2

20 − 10x− 15 cos(x+ 15) = 0.
Si ara dibuixem el gràfic de la part esquerra d’aquesta equació obtindrem un altre cop solucions en cada un
dels talls amb l’eix de les x.

> eqn:=x^2/20-10*x-15*cos(x+15)=0;

> plot(lhs(eqn),x=-10..10);

A partir del gràfic sembla que hi ha una solució a l’interval [1, 2].
Ara dirigirem Maple cap a trobar una solució en aquest interval.

> fsolve(eqn,x=1..2);

Hem trobat totes les solucions d’aquesta equació? De fet hi ha una altra solució! Per a trobar-la comenceu
estirant l’interval sobre el que heu fet el gràfic. Aleshores utilitzeu fsolve( ) per a obtenir una aproximació
numèrica d’aquesta segona solució.

Exercici 4.2

Determineu totes les solucions de l’equació x5 − 4x3 + 3x2 + 7x− 1 = 0. Comenceu mirant un gràfic signi-
ficatiu.

Exercici 4.3

Determineu totes les solucions de l’equació x2 − 2 = ln(x+ 5). Utilitzeu el gràfic d’una expressió per a
localitzar les solucions. Comproveu cada una de les solucions substituint-la en la’equació original.

Exercici 4.4

Els gràfics de y = 10− x2 i de y = 4 sin(2x) + 5 s’intersequen dues vegades sobre l’interval [−5, 5].
a) Feu el gràfic de les dues equacions juntes i estimeu amb el ratoĺı els punts d’intersecció.
b) Escriviu una equació que es pugui resoldre per a determinar les coordenades x dels puns d’intersecció.
c) Utilitzeu fsolve( ) per a resoldre aquesta equació.
d) Utilitzeu els resultats de la part c) per a estimar les coordenades y dels punts d’intersecció.
e) Sembla que les corbes es poden intersectar en un tercer punt a prop de (1, 9). Utilitzeu fsolve( ) i/o un
gràfic significatiu per a demostrar que no hi ha cap punt d’intersecció en aquest lloc.

4.4 Resoldre equacions formals

Sovint Maple pot resoldre equacions formals per a qualsevol de les variables.
Suposeu que volem obtenir la solució per a la variable g de l’equació: 4− v = 2T − k g .
La comanda solve( ) funciona bé en aquest cas.

> solve(4-v=2*T-k*g,g);

Aqúı hi ha una manera més maca per a mostrar el mateix resultat:
> g=solve(4-v=2*T-k*g,g);

Exercici 4.4

Editeu l’última comanda per a trobar solucions per a les altres lletres T, k i v.

Exercici 4.5

Resoleu l’equació x2 + y2 = 9 per a la variable y. Assigneu el conjunt de solucions a una variable que es digui
S. Quina relació hi ha entre les solucions S[1] i S[2]?

4.5 Resoldre sistemes d’equacions lineals utilitzant la comanda solve( )

Recordeu que cal reinicialitzar les variables abans de continuar. Carreguem també el paquet plots:
> restart:

> with(plots):

La comanda solve( ) també es pot utilitzar per a resoldre un sistema de m equacions lineals amb n
incògnites. En direm un sistema lineal m per n perquè sigui més curt.

Exemple 1:
Resoldre el sistema 2 per 2: 3x+ 2 y = 3 i x− y = −4

> solve({3*x+2*y=3,x-y=-4});
Un gràfic de les dues funcions subjacents mostra que la solució correspon al punt d’intersecció en (−1, 3).
Però primer necessitem obtenir la forma expĺıcita de cada una de les dues funcions lineals abans de poder
fer-ne el dibuix. Per tant resolem cada una de les equacions respecte y.

> y1:=solve(3*x+2*y=3,y);

> y2:=solve(x-y=-4,y);

Ara construim un gràfic format de dues parts: la “part1” conté els gràfics de les dues equacions i la “part2”
dibuixa el punt que és la solució que hem trobat. Aquest punt ha de ser el punt d’intersecció de les dues
ĺınies. Ho és efectivament?

> part1:=plot([y1,y2],x=-5..5):

> part2:=plot([[-1,3]],style=point,color=blue,symbol=circle):

> display([part1,part2]);

Exemple 2:
Aqúı hi ha un exemple de solució d’un sistema 3 per 3 amb incògnites x, y i z.
Resolem el sistema 3 per 3: {x+ y + z = 1, 3x+ y = 3, x− 2 y − z = 0}

> solve({x+y+z=1, 3*x+y=3, x-2*y-z=0});

Exercici 4.6

Determineu la solució del sistema:
4x+ 3 y = 12, 5x− 7 y = 35
Comproveu la solució substituint els valors que s’obtenen en les dues equacions del sistema.

Sistemes lineals amb un nombre de solucions infinit

Quan un sistema té més incògnites que equacions sovint trobem no una si no nombre infinit de solucions.
Aqúı hi ha un exemple.
Exemple 1:
Resolem el sistema: {x+ y + z = 1, 3x+ y = 3}.

> solns:=solve({x+y+z=1, 3*x+y=3});
Noteu que aquest cop no obtenim un únic conjunt valors numèrics per a x, y i z. En comptes d’això Maple
ens diu com han d’estar relacionats els valors de x, y i z per a construir una solució t́ıpica.
En particular l’expressió x = x en la resposta anterior indica que l’incògnita x pot ser qualsevol nombre. Ens
referirem a aquesta incògnita com la variable lliure de la solució. Per a obtenir qualsevol solució particular
(entre el nombre infinit que hi ha) trieu un valor per a la x i utilitzeu-lo per a calcular els valors corresponents
de la y i de la z. Per exemple si x = 4.

> subs(x=4,solns);

Aix́ı una solució és: x = 4 , y = −9 i z = 6.
Preneu-vos un minut de temps i verifiqueu a ma que aquests tres nombres satisfan realment les equacions
originals: x+ y + z = 1 i 3x+ y = 3.
Mirem ara la solució generada quan prenem x = 2.

> subs(x=2,solns);

Aix́ı que dues de les infinites solucions que hi ha són: (x, y, z) = (4, −9, 6) i (2, −3, 2).

Exercici 4.7

Resoleu el sistema: {x+ 2 y + z = 2, 3x+ y = 1} i doneu com a mı́nim tres solucions particulars.


